

NAG Library Routine Document

E05UCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

E05UCF is designed to find the global minimum of an arbitrary smooth function subject to constraints (which may include simple bounds on the variables, linear constraints and smooth nonlinear constraints) by generating a number of different starting points and performing a local search from each using sequential quadratic programming.

2 Specification

```

SUBROUTINE E05UCF (N, NCLIN, NCNLN, A, LDA, BL, BU, CONFUN, OBJFUN, NPTS,      &
                  X, LDX, START, REPEAT, NB, OBJF, OBJGRD, LDOBJD, ITER,      &
                  C, LDC, CJAC, LDCJAC, SDCJAC, R, LDR, SDR, CLAMDA,        &
                  LDCLDA, ISTATE, LISTAT, IOPTS, OPTS, IUSER, RUSER, INFO,  &
                  IFAIL)

INTEGER           N, NCLIN, NCNLN, LDA, NPTS, LDX, NB, LDOBJD, ITER(NB),    &
                  LDC, LDCJAC, SDCJAC, LDR, SDR, LDCLDA,                  &
                  ISTATE(LISTAT,NB), LISTAT, IOPTS(740), IUSER(*),        &
                  INFO(NB), IFAIL

REAL (KIND=nag_wp) A(LDA,*), BL(N+NCLIN+NCNLN), BU(N+NCLIN+NCNLN),      &
                  X(LDX,NB), OBJF(NB), OBJGRD(LDOBJD,NB), C(LDC,NB),      &
                  CJAC(LDCJAC,SDCJAC,NB), R(LDR,SDR,NB),                 &
                  CLAMDA(LDCLDA,NB), OPTS(485), RUSER(*)

LOGICAL          REPEAT

EXTERNAL        CONFUN, OBJFUN, START

```

Before calling E05UCF, the optional parameter arrays IOPTS and OPTS **must** be initialized for use with E05UCF by calling E05ZKF with OPTSTR set to 'Initialize = e05ucf'. Optional parameters may be specified by calling E05ZKF before the call to E05UCF.

3 Description

The problem is assumed to be stated in the following form:

$$\underset{x \in R^n}{\text{minimize}} F(x) \quad \text{subject to} \quad l \leq \begin{pmatrix} x \\ A_L x \\ c(x) \end{pmatrix} \leq u, \quad (1)$$

where $F(x)$ (the *objective function*) is a nonlinear function, A_L is an n_L by n linear constraint matrix, and $c(x)$ is an n_N element vector of nonlinear constraint functions. (The matrix A_L and the vector $c(x)$ may be empty.) The objective function and the constraint functions are assumed to be smooth, i.e., at least twice-continuously differentiable. (This routine will usually solve (1) if there are only isolated discontinuities away from the solution.)

E05UCF solves a user-specified number of local optimization problems with different starting points. You may specify the starting points via the subroutine START. If a random number generator is used to generate the starting points then the parameter REPEAT allows you to specify whether a repeatable set of points are generated or whether different starting points are generated on different calls. The resulting local minima are ordered and the best NB results returned in order of ascending values of the resulting objective function values at the minima. Thus the value returned in position 1 will be the best result obtained. If a sufficient number of different points are chosen then this is likely to be the global minimum. Please note that the default version of START uses a random number generator to generate the starting points.

4 References

- Dennis J E Jr and Moré J J (1977) Quasi-Newton methods, motivation and theory *SIAM Rev.* **19** 46–89
- Dennis J E Jr and Schnabel R B (1981) A new derivation of symmetric positive-definite secant updates *nonlinear programming* (eds O L Mangasarian, R R Meyer and S M Robinson) **4** 167–199 Academic Press
- Dennis J E Jr and Schnabel R B (1983) *Numerical Methods for Unconstrained Optimization and Nonlinear Equations* Prentice–Hall
- Fletcher R (1987) *Practical Methods of Optimization* (2nd Edition) Wiley
- Gill P E, Hammarling S, Murray W, Saunders M A and Wright M H (1986) Users' guide for LSSOL (Version 1.0) *Report SOL 86-1* Department of Operations Research, Stanford University
- Gill P E, Murray W, Saunders M A and Wright M H (1984) Users' guide for SOL/QPSOL version 3.2 *Report SOL 84-5* Department of Operations Research, Stanford University
- Gill P E, Murray W, Saunders M A and Wright M H (1986a) Some theoretical properties of an augmented Lagrangian merit function *Report SOL 86-6R* Department of Operations Research, Stanford University
- Gill P E, Murray W, Saunders M A and Wright M H (1986b) Users' guide for NPSOL (Version 4.0): a Fortran package for nonlinear programming *Report SOL 86-2* Department of Operations Research, Stanford University
- Gill P E, Murray W and Wright M H (1981) *Practical Optimization* Academic Press
- Powell M J D (1974) Introduction to constrained optimization *Numerical Methods for Constrained Optimization* (eds P E Gill and W Murray) 1–28 Academic Press
- Powell M J D (1983) Variable metric methods in constrained optimization *Mathematical Programming: the State of the Art* (eds A Bachem, M Grötschel and B Korte) 288–311 Springer–Verlag

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of variables.
Constraint: $N > 0$.
- 2: NCLIN – INTEGER *Input*
On entry: n_L , the number of general linear constraints.
Constraint: $NCLIN \geq 0$.
- 3: NCNLN – INTEGER *Input*
On entry: n_N , the number of nonlinear constraints.
Constraint: $NCNLN \geq 0$.
- 4: A(LDA,*) – REAL (KIND=nag_wp) array *Input*
Note: the second dimension of the array A must be at least N if $NCLIN > 0$, and at least 1 otherwise.
On entry: the matrix A_L of general linear constraints in (1). That is, the i th row contains the coefficients of the i th general linear constraint, for $i = 1, 2, \dots, NCLIN$.
 If $NCLIN = 0$, the array A is not referenced.

5: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which E05UCF is called.

Constraint: $LDA \geq NCLIN$.

6: BL(N + NCLIN + NCNLN) – REAL (KIND=nag_wp) array Input

7: BU(N + NCLIN + NCNLN) – REAL (KIND=nag_wp) array Input

On entry: BL must contain the lower bounds and BU the upper bounds for all the constraints in the following order. The first n elements of each array must contain the bounds on the variables, the next n_L elements the bounds for the general linear constraints (if any) and the next n_N elements the bounds for the general nonlinear constraints (if any). To specify a nonexistent lower bound (i.e., $l_j = -\infty$), set $BL(j) \leq -bigbnd$, and to specify a nonexistent upper bound (i.e., $u_j = +\infty$), set $BU(j) \geq bigbnd$; the default value of $bigbnd$ is 10^{20} , but this may be changed by the optional parameter **Infinite Bound Size**. To specify the j th constraint as an equality, set $BL(j) = BU(j) = \beta$, say, where $|\beta| < bigbnd$.

Constraints:

$BL(j) \leq BU(j)$, for $j = 1, 2, \dots, N + NCLIN + NCNLN$;
if $BL(j) = BU(j) = \beta$, $|\beta| < bigbnd$.

8: CONFUN – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

CONFUN must calculate the vector $c(x)$ of nonlinear constraint functions and (optionally) its Jacobian ($= \frac{\partial c}{\partial x}$) for a specified n -element vector x . If there are no nonlinear constraints (i.e., $NCNLN = 0$), CONFUN will never be called by E05UCF and CONFUN may be the dummy routine E04UDM. (E04UDM is included in the NAG Library.) If there are nonlinear constraints, the first call to CONFUN will occur before the first call to OBJFUN.

The specification of CONFUN is:

```
SUBROUTINE CONFUN (MODE, NCNLN, N, LDCJSL, NEEDC, X, C, CJSL,      &
                   NSTATE, IUSER, RUSER)
INTEGER              MODE, NCNLN, N, LDCJSL, NEEDC(NCNLN), NSTATE,      &
                   IUSER(*)
REAL (KIND=nag_wp) X(N), C(NCNLN), CJSL(LDCJSL,N), RUSER(*)
```

1: MODE – INTEGER Input/Output

On entry: indicates which values must be assigned during each call of CONFUN. Only the following values need be assigned, for each value of i such that $NEEDC(i) > 0$:

MODE = 0

$C(i)$.

MODE = 1

All available elements in the i th row of CJSL.

MODE = 2

$C(i)$ and all available elements in the i th row of CJSL.

On exit: may be set to a negative value if you wish to abandon the solution to the current local minimization problem. In this case E05UCF will move to the next local minimization problem.

2: NCNLN – INTEGER Input

On entry: n_N , the number of nonlinear constraints.

- | | | |
|----|--|---------------------|
| 3: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of variables. | |
| 4: | LDCJSL – INTEGER | <i>Input</i> |
| | <i>On entry:</i> LDCJSL is the same value as LDCJAC in the call to E05UCF. | |
| 5: | NEEDC(NCNLN) – INTEGER array | <i>Input</i> |
| | <i>On entry:</i> the indices of the elements of C and/or CJSL that must be evaluated by CONFUN. If $NEEDC(i) > 0$, $C(i)$ and/or the available elements of the i th row of CJSL (see parameter MODE) must be evaluated at x . | |
| 6: | X(N) – REAL (KIND=nag_wp) array | <i>Input</i> |
| | <i>On entry:</i> x , the vector of variables at which the constraint functions and/or the available elements of the constraint Jacobian are to be evaluated. | |
| 7: | C(NCNLN) – REAL (KIND=nag_wp) array | <i>Output</i> |
| | <i>On exit:</i> if $NEEDC(k) > 0$ and $MODE = 0$ or 2 , $C(k)$ must contain the value of $c_k(x)$. The remaining elements of C, corresponding to the non-positive elements of NEEDC, need not be set. | |
| 8: | CJSL(LDCJSL,N) – REAL (KIND=nag_wp) array | <i>Input/Output</i> |
| | CJSL may be regarded as a two-dimensional ‘slice’ of the three-dimensional array CJAC of E05UCF. | |

On entry: unless **Derivative Level** = 2 or 3 (the default setting is **Derivative Level** = 3, the elements of CJSL are set to special values which enable E05UCF to detect whether they are changed by CONFUN.

On exit: if $NEEDC(k) > 0$ and $MODE = 1$ or 2 , the k th row of CJSL must contain the available elements of the vector ∇c_k given by

$$\nabla c_k = \left(\frac{\partial c_k}{\partial x_1}, \frac{\partial c_k}{\partial x_2}, \dots, \frac{\partial c_k}{\partial x_n} \right)^T,$$

where $\frac{\partial c_k}{\partial x_j}$ is the partial derivative of the k th constraint with respect to the j th variable, evaluated at the point x . See also the parameter NSTATE. The remaining rows of CJSL, corresponding to non-positive elements of NEEDC, need not be set.

If all elements of the constraint Jacobian are known (i.e., **Derivative Level** = 2 or 3), any constant elements may be assigned to CJSL one time only at the start of each local optimization. An element of CJSL that is not subsequently assigned in CONFUN will retain its initial value throughout the local optimization. Constant elements may be loaded into CJSL during the first call to CONFUN for the local optimization (signalled by the value NSTATE = 1). The ability to preload constants is useful when many Jacobian elements are identically zero, in which case CJSL may be initialized to zero and nonzero elements may be reset by CONFUN.

Note that constant nonzero elements do affect the values of the constraints. Thus, if $CJSL(k, j)$ is set to a constant value, it need not be reset in subsequent calls to CONFUN, but the value $CJSL(k, j) \times X(j)$ must nonetheless be added to $C(k)$. For example, if $CJSL(1, 1) = 2$ and $CJSL(1, 2) = -5$ then the term $2 \times X(1) - 5 \times X(2)$ must be included in the definition of $C(1)$.

It must be emphasized that, if **Derivative Level** = 0 or 1, unassigned elements of CJSL are not treated as constant; they are estimated by finite differences, at nontrivial expense. If you do not supply a value for the optional parameter **Difference Interval**, an interval for each element of x is computed automatically at the start of each local optimization.

The automatic procedure can usually identify constant elements of CJSL, which are then computed once only by finite differences.

9: NSTATE – INTEGER *Input*

On entry: if NSTATE = 1 then E05UCF is calling CONFUN for the first time on the current local optimization problem. This parameter setting allows you to save computation time if certain data must be calculated only once.

10: IUSER(*) – INTEGER array *User Workspace*

11: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*

CONFUN is called with the parameters IUSER and RUSER as supplied to E05UCF. You are free to use the arrays IUSER and RUSER to supply information to CONFUN as an alternative to using COMMON global variables.

CONFUN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which E05UCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

CONFUN should be tested separately before being used in conjunction with E05UCF. See also the description of the optional parameter **Verify**.

9: OBJFUN – SUBROUTINE, supplied by the user. *External Procedure*

OBJFUN must calculate the objective function $F(x)$ and (optionally) its gradient $g(x) = \frac{\partial F}{\partial x}$ for a specified n -vector x .

The specification of OBJFUN is:

```
SUBROUTINE OBJFUN (MODE, N, X, OBJF, OBJGRD, NSTATE, IUSER, RUSER)
```

```
INTEGER MODE, N, NSTATE, IUSER(*)
```

```
REAL (KIND=nag_wp) X(N), OBJF, OBJGRD(N), RUSER(*)
```

1: MODE – INTEGER *Input/Output*

On entry: indicates which values must be assigned during each call of OBJFUN. Only the following values need be assigned:

MODE = 0

OBJF.

MODE = 1

All available elements of OBJGRD.

MODE = 2

OBJF and all available elements of OBJGRD.

On exit: may be set to a negative value if you wish to abandon the solution to the current local minimization problem. In this case E05UCF will move to the next local minimization problem.

2: N – INTEGER *Input*

On entry: n , the number of variables.

3: X(N) – REAL (KIND=nag_wp) array *Input*

On entry: x , the vector of variables at which the objective function and/or all available elements of its gradient are to be evaluated.

4:	OBJF – REAL (KIND=nag_wp)	<i>Output</i>
	<i>On exit:</i> if MODE = 0 or 2, OBJF must be set to the value of the objective function at x .	
5:	OBJGRD(N) – REAL (KIND=nag_wp) array	<i>Input/Output</i>
	<i>On entry:</i> the elements of OBJGRD are set to special values which enable E05UCF to detect whether they are changed by OBJFUN.	
	<i>On exit:</i> if MODE = 1 or 2, OBJGRD must return the available elements of the gradient evaluated at x .	
6:	NSTATE – INTEGER	<i>Input</i>
	<i>On entry:</i> if NSTATE = 1 then E05UCF is calling OBJFUN for the first time on the current local optimization problem. This parameter setting allows you to save computation time if certain data must be calculated only once.	
7:	IUSER(*) – INTEGER array	<i>User Workspace</i>
8:	RUSER(*) – REAL (KIND=nag_wp) array	<i>User Workspace</i>
	OBJFUN is called with the parameters IUSER and RUSER as supplied to E05UCF. You are free to use the arrays IUSER and RUSER to supply information to OBJFUN as an alternative to using COMMON global variables.	

OBJFUN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which E05UCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

OBJFUN should be tested separately before being used in conjunction with E05UCF. See also the description of the optional parameter **Verify**.

- 10: NPTS – INTEGER *Input*
- On entry:* the number of different starting points to be generated and used. The more points used, the more likely that the best returned solution will be a global minimum.
- Constraint:* $1 \leq \text{NB} \leq \text{NPTS}$.
- 11: X(LDX,NB) – REAL (KIND=nag_wp) array *Output*
- On exit:* $X(j,i)$ contains the final estimate of the i th solution, for $j = 1, 2, \dots, \text{N}$.
- 12: LDX – INTEGER *Input*
- On entry:* the first dimension of the array X as declared in the (sub)program from which E05UCF is called.
- Constraint:* $\text{LDX} \geq \text{N}$.
- 13: START – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*
- START must calculate the NPTS starting points to be used by the local optimizer. If you do not wish to write a routine specific to your problem then E05UCZ may be used as the actual argument. E05UCZ is supplied in the NAG Library and uses the NAG quasi-random number generators to distribute starting points uniformly across the domain. It is affected by the value of REPEAT.

The specification of START is:

```
SUBROUTINE START (NPTS, QUAS, N, REPEAT, BL, BU, IUSER, RUSER, MODE)
  INTEGER          NPTS, N, IUSER(*), MODE
  REAL (KIND=nag_wp) QUAS(N,NPTS), BL(N), BU(N), RUSER(*)
  LOGICAL          REPEAT
```

1:	NPTS – INTEGER	<i>Input</i>
	<i>On entry:</i> indicates the number of starting points.	
2:	QUAS(N,NPTS) – REAL (KIND=nag_wp) array	<i>Input/Output</i>
	<i>On entry:</i> all elements of QUAS will have been set to zero, so only nonzero values need be set subsequently.	
	<i>On exit:</i> must contain the starting points for the NPTS local minimizations, i.e., QUAS(<i>j</i> , <i>i</i>) must contain the <i>j</i> th component of the <i>i</i> th starting point.	
3:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of variables.	
4:	REPEAT – LOGICAL	<i>Input</i>
	<i>On entry:</i> specifies whether a repeatable or non-repeatable sequence of points are to be generated.	
5:	BL(N) – REAL (KIND=nag_wp) array	<i>Input</i>
	<i>On entry:</i> the lower bounds on the variables. These may be used to ensure that the starting points generated in some sense ‘cover’ the region, but there is no requirement that a starting point be feasible.	
6:	BU(N) – REAL (KIND=nag_wp) array	<i>Input</i>
	<i>On entry:</i> the upper bounds on the variables. (See BL.)	
7:	IUSER(*) – INTEGER array	<i>User Workspace</i>
8:	RUSER(*) – REAL (KIND=nag_wp) array	<i>User Workspace</i>
	START is called with the parameters IUSER and RUSER as supplied to E05UCF. You are free to use the arrays IUSER and RUSER to supply information to START as an alternative to using COMMON global variables.	
9:	MODE – INTEGER	<i>Input/Output</i>
	<i>On entry:</i> MODE will contain 0.	
	<i>On exit:</i> if you set MODE to a negative value then E05UCF will terminate immediately with IFAIL = 9.	

START must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which E05UCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 14: REPEAT – LOGICAL *Input*
On entry: is passed as an argument to START and may be used to initialize a random number generator to a repeatable, or non-repeatable, sequence.
- 15: NB – INTEGER *Input*
On entry: the number of solutions to be returned. The routine saves up to NB local minima ordered by increasing value of the final objective function. If the defining criterion for ‘best solution’ is only that the value of the objective function is as small as possible then NB should be set to 1. However, if you want to look at other solutions that may have desirable properties then setting NB > 1 will produce NB local minima, ordered by increasing value of their objective functions at the minima.
Constraint: $1 \leq \text{NB} \leq \text{NPTS}$.

- 16: OBJF(NB) – REAL (KIND=nag_wp) array *Output*
On exit: OBJF(i) contains the value of the objective function at the final iterate for the i th solution.
- 17: OBJGRD(LDOBJD,NB) – REAL (KIND=nag_wp) array *Output*
On exit: OBJGRD(j, i) contains the gradient of the objective function for the i th solution at the final iterate (or its finite difference approximation), for $j = 1, 2, \dots, N$.
- 18: LDOBJD – INTEGER *Input*
On entry: the first dimension of the array OBJGRD as declared in the (sub)program from which E05UCF is called.
Constraint: LDOBJD \geq N.
- 19: ITER(NB) – INTEGER array *Output*
On exit: ITER(i) contains the number of major iterations performed to obtain the i th solution. If less than NB solutions are returned then ITER(NB) contains the number of starting points that have resulted in a converged solution. If this is close to NPTS then this might be indicative that fewer than NB local minima exist.
- 20: C(LDC,NB) – REAL (KIND=nag_wp) array *Output*
On exit: if NCNLN $>$ 0, C(j, i) contains the value of the j th nonlinear constraint function c_j at the final iterate, for the i th solution, for $j = 1, 2, \dots, \text{NCNLN}$.
 If NCNLN = 0, the array C is not referenced.
- 21: LDC – INTEGER *Input*
On entry: the first dimension of the array C as declared in the (sub)program from which E05UCF is called.
Constraint: LDC \geq NCNLN.
- 22: CJAC(LDCJAC,SDCJAC,NB) – REAL (KIND=nag_wp) array *Output*
On exit: if NCNLN $>$ 0, CJAC contains the Jacobian matrices of the nonlinear constraint functions at the final iterate for each of the returned solutions, i.e., CJAC(k, j, i) contains the partial derivative of the k th constraint function with respect to the j th variable, for $k = 1, 2, \dots, \text{NCNLN}$ and $j = 1, 2, \dots, N$, for the i th solution. (See the discussion of parameter CJSL under CONFUN.)
 If NCNLN = 0, the array CJAC is not referenced.
- 23: LDCJAC – INTEGER *Input*
On entry: the first dimension of the array CJAC as declared in the (sub)program from which E05UCF is called.
Constraint: LDCJAC \geq NCNLN.
- 24: SDCJAC – INTEGER *Input*
On entry: the second dimension of the array CJAC as declared in the (sub)program from which E05UCF is called.
Constraint: if NCNLN $>$ 0, SDCJAC \geq N.
- 25: R(LDR,SDR,NB) – REAL (KIND=nag_wp) array *Output*
On exit: for each of the NB solutions R will contain a form of the Hessian; for the i th returned solution R(LDR, SDR, i) contains the Hessian that would be returned from the local minimizer. If **Hessian** = NO, the default, each R(LDR, SDR, i) contains the upper triangular Cholesky factor R of $Q^T H Q$, an estimate of the transformed and reordered Hessian of the Lagrangian at x . If

Hessian = YES, R(LDR,SDR,*i*) contains the upper triangular Cholesky factor *R* of *H*, the approximate (untransformed) Hessian of the Lagrangian, with the variables in the natural order.

26: LDR – INTEGER *Input*

On entry: the first dimension of the array R as declared in the (sub)program from which E05UCF is called.

Constraint: LDR ≥ N.

27: SDR – INTEGER *Input*

On entry: the second dimension of the array R as declared in the (sub)program from which E05UCF is called.

Constraint: SDR ≥ N.

28: CLAMDA(LDCLDA,NB) – REAL (KIND=nag_wp) array *Output*

On exit: the values of the QP multipliers from the last QP subproblem solved for the *i*th solution. CLAMDA(*j*,*i*) should be non-negative if ISTATE(*j*,*i*) = 1 and non-positive if ISTATE(*j*,*i*) = 2.

29: LDCLDA – INTEGER *Input*

On entry: the first dimension of the array CLAMDA as declared in the (sub)program from which E05UCF is called.

Constraint: LDCLDA ≥ N + NCLIN + NCNLN.

30: ISTATE(LISTAT,NB) – INTEGER array *Output*

On exit: ISTATE(*j*,*i*) contains the status of the constraints in the QP working set for the *i*th solution. The significance of each possible value of ISTATE(*j*,*i*) is as follows:

ISTATE(<i>j</i> , <i>i</i>)	Meaning
0	The constraint is satisfied to within the feasibility tolerance, but is not in the QP working set.
1	This inequality constraint is included in the QP working set at its lower bound.
2	This inequality constraint is included in the QP working set at its upper bound.
3	This constraint is included in the QP working set as an equality. This value of ISTATE can occur only when BL(<i>j</i>) = BU(<i>j</i>).

31: LISTAT – INTEGER *Input*

On entry: the first dimension of the array ISTATE as declared in the (sub)program from which E05UCF is called.

Constraint: LISTAT ≥ N + NCLIN + NCNLN.

32: IOPTS(740) – INTEGER array *Communication Array*

33: OPTS(485) – REAL (KIND=nag_wp) array *Communication Array*

The arrays IOPTS and OPTS **must not** be altered between calls to any of the routines E05UCF and E05ZKF.

34: IUSER(*) – INTEGER array *User Workspace*

35: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*

IUSER and RUSER are not used by E05UCF, but are passed directly to CONFUN, OBJFUN and START and may be used to pass information to these routines as an alternative to using COMMON global variables.

With care, you may also write information back into IUSER and RUSER. This might be useful, for example, should there be a need to preserve the state of a random number generator.

With SMP-enabled versions of E05UCF the arrays IUSER and RUSER provided are classified as OpenMP shared memory. Use of IUSER and RUSER has to take account of this in order to preserve thread safety whenever information is written back to either of these arrays.

36: INFO(NB) – INTEGER array *Output*

On exit: INFO(i) contains one of 0, 1 or 6. Please see the description of each corresponding value of IFAIL on exit from E04UCF/E04UCA for detailed explanations of these exit values. As usual 0 denotes success.

If IFAIL = 8 on exit, then not all NB solutions have been found, and INFO(NB) contains the number of solutions actually found.

37: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL \neq 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: E05UCF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

An input value is incorrect. One or more of the following constraints are violated.

On entry, $BL(i) > BU(i)$: $i = \langle value \rangle$.

Constraint: $BL(i) \leq BU(i)$, for all i .

On entry, $LDA = \langle value \rangle$ and $NCLIN = \langle value \rangle$.

Constraint: $LDA \geq NCLIN$.

On entry, $LDC = \langle value \rangle$ and $NCNLN = \langle value \rangle$.

Constraint: $LDC \geq NCNLN$.

On entry, $LDCJAC = \langle value \rangle$ and $NCNLN = \langle value \rangle$.

Constraint: $LDCJAC \geq NCNLN$.

On entry, $LDCLDA = \langle value \rangle$, $N = \langle value \rangle$, $NCLIN = \langle value \rangle$ and $NCNLN = \langle value \rangle$.

Constraint: $LDCLDA \geq N + NCLIN + NCNLN$.

On entry, $LDOBJD = \langle value \rangle$ and $N = \langle value \rangle$.

Constraint: $LDOBJD \geq N$.

On entry, $LDR = \langle value \rangle$ and $N = \langle value \rangle$.

Constraint: $LDR \geq N$.

On entry, $LDX = \langle value \rangle$ and $N = \langle value \rangle$.

Constraint: $LDX \geq N$.

On entry, $LISTAT = \langle value \rangle$, $N = \langle value \rangle$, $NCLIN = \langle value \rangle$ and $NCNLN = \langle value \rangle$.

Constraint: $LISTAT \geq N + NCLIN + NCNLN$.

On entry, $N = \langle value \rangle$.

Constraint: $N > 0$.

On entry, $NB = \langle value \rangle$ and $NPTS = \langle value \rangle$.

Constraint: $1 \leq NB \leq NPTS$.

On entry, $NCLIN = \langle value \rangle$.

Constraint: $NCLIN \geq 0$.

On entry, $NCNLN = \langle value \rangle$.

Constraint: $NCNLN \geq 0$.

On entry, $NCNLN > 0$, $SDCJAC = \langle value \rangle$ and $N = \langle value \rangle$.

Constraint: if $NCNLN > 0$, $SDCJAC \geq N$.

On entry, $SDR = \langle value \rangle$ and $N = \langle value \rangle$.

Constraint: $SDR \geq N$.

IFAIL = 2

No solution obtained. Linear constraints may be infeasible.

E05UCF has terminated without finding any solutions. The majority of calls to the local optimizer have failed to find a feasible point for the linear constraints and bounds, which means that either no feasible point exists for the given value of the optional parameter **Linear Feasibility Tolerance** (default value $\sqrt{\epsilon}$, where ϵ is the *machine precision*), or no feasible point could be found in the number of iterations specified by the optional parameter **Minor Iteration Limit**. You should check that there are no constraint redundancies. If the data for the constraints are accurate only to an absolute precision σ , you should ensure that the value of the optional parameter **Linear Feasibility Tolerance** is greater than σ . For example, if all elements of A_L are of order unity and are accurate to only three decimal places, **Linear Feasibility Tolerance** should be at least 10^{-3} .

IFAIL = 3

E05UCF has failed to find any solutions. The majority of local optimizations could not find a feasible point for the nonlinear constraints. The problem may have no feasible solution. This behaviour will occur if there is no feasible point for the nonlinear constraints. (However, there is no general test that can determine whether a feasible point exists for a set of nonlinear constraints.)

No solution obtained. Nonlinear constraints may be infeasible.

IFAIL = 4

No solution obtained. Many potential solutions reach iteration limit.

The **Iteration Limit** may be changed using E05ZKF.

IFAIL = 7

User-supplied derivatives probably wrong. The user-supplied derivatives of the objective function and/or nonlinear constraints appear to be incorrect.

Large errors were found in the derivatives of the objective function and/or nonlinear constraints. This value of IFAIL will occur if the verification process indicated that at least one gradient or Jacobian element had no correct figures. You should refer to or enable the printed output to determine which elements are suspected to be in error.

As a first-step, you should check that the code for the objective and constraint values is correct – for example, by computing the function at a point where the correct value is known. However, care should be taken that the chosen point fully tests the evaluation of the function. It is remarkable how often the values $x = 0$ or $x = 1$ are used to test function evaluation procedures, and how often the special properties of these numbers make the test meaningless.

Gradient checking will be ineffective if the objective function uses information computed by the constraints, since they are not necessarily computed before each function evaluation.

Errors in programming the function may be quite subtle in that the function value is ‘almost’ correct. For example, the function may not be accurate to full precision because of the inaccurate calculation of a subsidiary quantity, or the limited accuracy of data upon which the function depends. A common error on machines where numerical calculations are usually performed in double precision is to include even one single precision constant in the calculation of the function; since some compilers do not convert such constants to double precision, half the correct figures may be lost by such a seemingly trivial error.

IFAIL = 8

Only *<value>* solutions obtained.

Not all NB solutions have been found. INFO(NB) contains the number actually found.

IFAIL = 9

User terminated computation from START procedure: MODE = *<value>*.

If E05UCZ has been used as an actual argument for START then the message displayed, when IFAIL = 0 or -1 on entry to E05UCF, will have the following meaning:

998 failure to allocate space, a smaller value of NPTS should be tried.

997 an internal error has occurred. Please contact NAG for assistance.

IFAIL = 10

Failed to initialize optional parameter arrays.

IFAIL = -999

Dynamic memory allocation failed.

7 Accuracy

If IFAIL = 0 on exit and the value of INFO(*i*) = 0, then the vector returned in the array X for solution *i* is an estimate of the solution to an accuracy of approximately **Optimality Tolerance**.

8 Further Comments

You should be wary of requesting much intermediate output from the local optimizer, since large volumes may be produced if NPTS is large.

The auxiliary routine E05UCZ makes use of the NAG quasi-random Sobol generator (G05YLF and G05YMF). If E05UCZ is used as an argument for START (see the description of START) and REPEAT = .FALSE. then a randomly chosen value for ISKIP is used, otherwise ISKIP is set to 100. If REPEAT is set to .FALSE. and the program is executed several times, each time producing the same best answer, then there is increased probability that this answer is a global minimum. However, if it is important that identical results be obtained on successive runs, then REPEAT should be set to .TRUE..

8.1 Description of the Printed Output

See Section 8.1 in E04UCF/E04UCA.

9 Example

This example finds the global minimum of the two-dimensional Schwefel function:

$$\underset{\mathbf{x} \in \mathbb{R}^2}{\text{minimize}} f = \sum_{j=1}^2 x_j \sin\left(\sqrt{|x_j|}\right)$$

subject to the constraints:

$$\begin{aligned}
 & -10000 < 3.0x_1 - 2.0x_2 < 10.0, \\
 & -1.0 < x_1^2 - x_2^2 + 3.0x_1x_2 < 500000.0, \\
 & -0.9 < \cos\left(\left(x_1/200\right)^2 + \left(x_2/100\right)\right) < 0.9, \\
 & -500 \leq x_1 \leq 500, \\
 & -500 \leq x_2 \leq 500.
 \end{aligned}$$

9.1 Program Text

```

!   Mark 24 Release.   NAG Copyright 2012.

Module e05ucfe_mod

!   E05UCF Example Program Module:
!   Parameters and User-defined Routines

!   .. Use Statements ..
Use nag_library, Only: nag_wp
!   .. Implicit None Statement ..
Implicit None
Contains
Subroutine schwefel_obj(mode,n,x,objf,objgrd,nstate,iuser,ruser)

!   .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out)      :: objf
Integer, Intent (Inout)               :: mode
Integer, Intent (In)                  :: n, nstate
!   .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout)    :: objgrd(n), ruser(*)
Real (Kind=nag_wp), Intent (In)       :: x(n)
Integer, Intent (Inout)               :: iuser(*)
!   .. Intrinsic Procedures ..
Intrinsic                              :: abs, cos, sin, sqrt, sum
!   .. Executable Statements ..
If (mode==0 .Or. mode==2) Then
!   Evaluate the objective function.
  objf = sum(x(1:n)*sin(sqrt(abs(x(1:n))))))
End If

If (mode==1 .Or. mode==2) Then
!   Calculate the gradient of the objective function.
  objgrd(1:n) = sin(sqrt(abs(x(1:n)))) + 0.5_nag_wp*sqrt(abs(x(1:n)))* &
    cos(sqrt(abs(x(1:n))))
End If

Return
End Subroutine schwefel_obj
Subroutine schwefel_confun(mode,ncnln,n,ldcjsl,needc,x,c,cjsl,nstate, &
  iuser,ruser)

!   .. Scalar Arguments ..
Integer, Intent (In)                  :: ldcjsl, n, ncnln, nstate
Integer, Intent (Inout)               :: mode
!   .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out)      :: c(ncnln)
Real (Kind=nag_wp), Intent (Inout)    :: cjsl(ldcjsl,n), ruser(*)
Real (Kind=nag_wp), Intent (In)       :: x(n)
Integer, Intent (Inout)               :: iuser(*)
Integer, Intent (In)                  :: needc(ncnln)
!   .. Local Scalars ..
Real (Kind=nag_wp)                   :: t1, t2
Integer                                :: k
!   .. Intrinsic Procedures ..
Intrinsic                              :: cos, sin
!   .. Executable Statements ..
If (mode==0 .Or. mode==2) Then
!   Constraint values are required.

```

```

!       Only those for which needc is non-zero need be set.
      Do k = 1, ncnln
        If (needc(k)>0) Then
          Select Case (k)
            Case (1)
              c(k) = x(1)**2 - x(2)**2 + 3.0_nag_wp*x(1)*x(2)
            Case (2)
              c(k) = cos((x(1)/200.0_nag_wp)**2+(x(2)/100.0_nag_wp))
          End Select
        End If
      End Do
    End If

    If (mode==1 .Or. mode==2) Then
!       Constraint derivatives are required.
      Do k = 1, ncnln
        Select Case (k)
          Case (1)
            cjsl(k,1) = 2.0_nag_wp*x(1) + 3.0_nag_wp*x(2)
            cjsl(k,2) = -2.0_nag_wp*x(2) + 3.0_nag_wp*x(1)
          Case (2)
            t1 = x(1)/200.0_nag_wp
            t2 = x(2)/100.0_nag_wp
            cjsl(k,1) = -sin(t1**2+t2)*2.0_nag_wp*t1/200.0_nag_wp
            cjsl(k,2) = -sin(t1**2+t2)/100.0_nag_wp
          End Select
        End Do
      End If

      Return
    End Subroutine schwefel_confun
  Subroutine mystart(npts,quas,n,repeat,bl,bu,iuser,ruser,mode)
!       Sets an initial point.
!       A typical user-defined start procedure (not used here)

!       .. Scalar Arguments ..
      Integer, Intent (Inout)           :: mode
      Integer, Intent (In)              :: n, npts
      Logical, Intent (In)              :: repeat
!       .. Array Arguments ..
      Real (Kind=nag_wp), Intent (In)   :: bl(n), bu(n)
      Real (Kind=nag_wp), Intent (Inout) :: quas(n,npts), ruser(*)
      Integer, Intent (Inout)          :: iuser(*)
!       .. Executable Statements ..
!       quas(1:n,1:npts) is pre-assigned to zero.
      quas(1,1) = 420.9687_nag_wp
      quas(2,2) = 1.0_nag_wp !etc
      Return
    End Subroutine mystart
  End Module e05ucfe_mod
  Program e05ucfe

!       E05UCF Example Main Program

!       .. Use Statements ..
      Use nag_library, Only: dgemv, e05ucf, e05ucz, e05z kf, nag_wp
      Use e05ucfe_mod, Only: schwefel_confun, schwefel_obj
!       .. Implicit None Statement ..
      Implicit None
!       .. Parameters ..
      Integer, Parameter                :: n = 2, nclin = 1, ncnln = 2, &
                                         nin = 5, nout = 6
!       .. Local Scalars ..
      Integer                           :: i, ifail, j, k, l, lda, ldc, &
                                         ldcjac, ldclda, ldobjd, ldr, &
                                         ldx, liopts, listat, lopts, nb, &
                                         npts, sda, sdcjac, sdr
      Logical                           :: repeat
!       .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,,:), bl(:), bu(:), c(:,,:), &
                                         cjac(:,,:), clamda(:,,:), &

```

```

                                objf(:), objgrd(:,,:), opts(:), &
                                r(:, :, :), work(:), x(:, :)
Real (Kind=nag_wp)                :: ruser(1)
Integer, Allocatable              :: info(:), iopts(:), istate(:, :), &
                                iter(:)
Integer                            :: iuser(1)
! .. Executable Statements ..
Write (nout,*) 'E05UCF Example Program Results'
Flush (nout)

! Skip heading in data file
Read (nin,*)

Read (nin,*) nb, npts
Read (nin,*) repeat

lda = nclin

If (nclin>0) Then
  sda = n
Else
  sda = 1
End If

ldx = n
ldobjd = n
ldc = ncnln
ldcjac = ncnln

If (ncnln>0) Then
  sdcjac = n
Else
  sdcjac = 0
End If

ldr = n
sdr = n
ldclda = n + nclin + ncnln
listat = n + nclin + ncnln
liopts = 740
lopts = 485
Allocate (a(lda,sda),bl(n+nclin+ncnln),bu(n+nclin+ncnln),x(ldx,nb), &
  objf(nb),objgrd(ldobjd,nb),iter(nb),c(ldc,nb),cjac(ldcjac,sdcjac,nb), &
  r(ldr,sdr,nb),clamda(ldclda,nb),istate(listat,nb),iopts(liopts), &
  opts(lopts),info(nb),work(nclin))

bl(1:n+nclin+ncnln) = (/ -500.0_nag_wp, -500.0_nag_wp, -10000.0_nag_wp, &
  -1.0_nag_wp, -0.9_nag_wp /)
bu(1:n+nclin+ncnln) = (/ 500.0_nag_wp, 500.0_nag_wp, 10.0_nag_wp, &
  500000.0_nag_wp, 0.9_nag_wp /)

a(1,1) = 3.0_nag_wp
a(1,2) = -2.0_nag_wp

! Initialize the solver.

ifail = 0
Call e05zkgf('Initialize = E05UCF',iopts,liopts,opts,lopts,ifail)

! Solve the problem.
! USE mystart from e05ucfe_mod and pass this instead of e05ucz to e05ucf
! to set your own start points.

ifail = -1
Call e05ucf(n,nclin,ncnln,a,lda,bl,bu,schwefel_confun,schwefel_obj,npts, &
  x,ldx,e05ucz,repeat,nb,objf,objgrd,ldobjd,iter,c,ldc,cjac,ldcjac, &
  sdcjac,r,ldr,sdr,clamda,ldclda,istate,listat,iopts,opts,iuser,ruser, &
  info,ifail)

Select Case (ifail)
Case (0)

```

```

    l = nb
  Case (8)
    l = info(nb)
    Write (nout,'(1X,I16,A)') iter(nb), ' starting points converged'
  Case Default
    Go To 100
  End Select

loop: Do i = 1, l
  Write (nout,99999) i
  Write (nout,99998) info(i)
  Write (nout,99997) 'Varbl'
  Do j = 1, n
    Write (nout,99996) 'V', j, istrate(j,i), x(j,i), clamda(j,i)
  End Do
  If (nclin>0) Then
    Write (nout,99997) 'L Con'

!       Below is a call to the level 2 BLAS routine DGEMV.
!       This performs the matrix vector multiplication A*X
!       (linear constraint values) and puts the result in
!       the first NCLIN locations of WORK.

    Call dgemv('N',nclin,n,1.0_nag_wp,a,lda,x(1,i),1,0.0_nag_wp,work,1)

    Do k = n + 1, n + nclin
      j = k - n
      Write (nout,99996) 'L', j, istrate(k,i), work(j), clamda(k,i)
    End Do
  End If
  If (ncnln>0) Then
    Write (nout,99997) 'N Con'
    Do k = n + nclin + 1, n + nclin + ncnln
      j = k - n - nclin
      Write (nout,99996) 'N', j, istrate(k,i), c(j,i), clamda(k,i)
    End Do
  End If
  Write (nout,99995) objf(i)
  Write (nout,99994)
  Write (nout,99993)(clamda(k,i),k=1,n+nclin+ncnln)

  If (l==1) Then
    Exit loop
  End If

  Write (nout,*)

  Write (nout,*) &
  '-----'

End Do loop

100  Continue

99999 Format (/1X,'Solution number',I16)
99998 Format (/1X,'Local minimization exited with code',I5)
99997 Format (/1X,A,2X,'Istate',3X,'Value',9X,'Lagr Mult'/)
99996 Format (1X,A,2(1X,I3),4X,F12.4,2X,F12.4)
99995 Format (/1X,'Final objective value = ',1X,F12.4)
99994 Format (/1X,'QP multipliers')
99993 Format (1X,F12.4)
      End Program e05ucfe

```

9.2 Program Data

```

E05UCF Example Program Data
2 1000          : NB, NPTS
T              : REPEAT

```


9.3 Program Results

E05UCF Example Program Results

```

Solution number          1
Local minimization exited with code    0

Varbl  Istate  Value      Lagr Mult
V   1   0      -394.1514    0.0000
V   2   0      -433.4910    0.0000

L Con  Istate  Value      Lagr Mult
L   1   0      -315.4722    0.0000

N Con  Istate  Value      Lagr Mult
N   1   0      480024.1069    0.0000
N   2   2           0.9000   -718.9449

Final objective value =    -731.7064

QP multipliers
    0.0000
    0.0000
    0.0000
    0.0000
   -718.9449

```

```

-----
Solution number          2
Local minimization exited with code    0

Varbl  Istate  Value      Lagr Mult
V   1   0      -413.8051    0.0000
V   2   0      -382.9839    0.0000

L Con  Istate  Value      Lagr Mult
L   1   0      -475.4474    0.0000

N Con  Istate  Value      Lagr Mult
N   1   2      500000.0000   -0.0062
N   2   2           0.9000  -1161.4993

Final objective value =    -665.1962

QP multipliers
    0.0000
    0.0000
    0.0000
   -0.0062
  -1161.4993

```

10 Algorithmic Details

See Section 10 in E04UCF/E04UCA.

11 Optional Parameters

Several optional parameters in E05UCF define choices in the problem specification or the algorithm logic. In order to reduce the number of formal parameters of E05UCF these optional parameters have associated *default values* that are appropriate for most problems. Therefore you need only specify those optional parameters whose values are to be different from their default values.

Optional parameters may be specified by calling E05ZKF before a call to E05UCF. Before calling E05UCF, the optional parameter arrays IOPTS and OPTS **must** be initialized for use with E05UCF by calling E05ZKF with OPTSTR set to 'Initialize = e05ucf'.

All optional parameters not specified are set to their default values. Optional parameters specified are unaltered by E05UCF (unless they define invalid values) and so remain in effect for subsequent calls to E05UCF.

11.1 Description of the Optional Parameters

E05UCF supports two options that are distinct from those of E04UCF/E04UCA:

Punch Unit *i* Default = 6

This option allows the user to send information arising from an appropriate setting of **Out_Level** to be sent to the Fortran unit number defined by **Punch Unit**. If the user wishes this file to be different to the standard output unit (6) where other output is displayed then this file should be attached by calling X04ACF prior to calling E05UCF.

Out_Level *i* Default = 0

This option defines the amount of extra information to be sent to the Fortran unit number defined by **Punch Unit**. The possible choices for *i* are the following:

<i>i</i>	Meaning
0	No extra output.
1	Updated solutions only. This is useful during long runs to observe progress.
2	Successful start points only. This is useful to save the starting points that gave rise to the final solution.
3	Both updated solutions and successful start points.

See Section 11 in E04UCF/E04UCA for details of the other options.

The **Warm Start** option of E04UCF/E04UCA is not a valid option for use with E05UCF.

12 Description of Monitoring Information

See Section 12 in E04UCF/E04UCA.
